@PACKAGE_INIT@

set_and_check(CTKAppLauncherLib_CMAKE_DIR "@PACKAGE_CMAKE_DIR_CONFIG@")
set_and_check(CTKAppLauncherLib_CONFIG_DIR "@PACKAGE_CONFIG_DIR_CONFIG@")
set_and_check(CTKAppLauncherLib_TARGETS "${CTKAppLauncherLib_CMAKE_DIR}/CTKAppLauncherLibTargets.cmake")

if(NOT CTKAppLauncherLib_TARGETS_IMPORTED)
  set(CTKAppLauncherLib_TARGETS_IMPORTED 1)
  include(${CTKAppLauncherLib_TARGETS})
endif()

set(CTKAppLauncherLib_LIBRARIES "@CTKAppLauncherLib_TARGETS@")

@CTKAppLauncher_CONFIG_CODE@

if(CTKAppLauncher_QT_STATIC_LIBRARIES)
  list(APPEND CTKAppLauncherLib_LIBRARIES ${CTKAppLauncher_QT_STATIC_LIBRARIES})
endif()

# Set expected list of Qt components
set(CTKAppLauncherLib_QT_VERSION "@CTKAppLauncherLib_QT_VERSION_CONFIG@")
if(CTKAppLauncherLib_QT_VERSION VERSION_GREATER "4")
  set(qt_components COMPONENTS "Widgets")
endif()

# Propagate QUIET and REQUIRED arguments
set(cmake_fd_quiet_arg)
if(${CMAKE_FIND_PACKAGE_NAME}_FIND_QUIETLY)
  set(cmake_fd_quiet_arg QUIET)
endif()
set(cmake_fd_required_arg)
if(${CMAKE_FIND_PACKAGE_NAME}_FIND_REQUIRED)
  set(cmake_fd_required_arg REQUIRED)
endif()

find_package(
  Qt${CTKAppLauncherLib_QT_VERSION}
  ${qt_components}
  ${cmake_fd_required_arg} ${cmake_fd_quiet_arg}
  )
